© 1982 Geologie en Mijnbouw 0016-7746/82/6103-0293 $ 2.50/0

SHORT COMMUNICATIONS

TEMPERATURE LOG ANALYSIS WITH A POCKET CALCULATOR: INTERPRETATION PROGRAM FOR DISCRETE TEMPERATURE LOGS IN SMALL-DIAMETER WELLS[1]

BEN KEET[2]

ABSTRACT

Keet, B. 1982 Temperature log analysis with a pocket calculator: Interpretation program for discrete temperature logs in small-diameter wells – Geol. Mijnbouw 61: 293-295

A program is presented to obtain a fine-scale correlation between discrete temperature logs and lithology. The algorithm is essentially the least square method. The limitations and an example of application of this program are given.

INTRODUCTION

The time consuming temperature log plotting, and the subjective interpretation of bends in those curves can be supplanted by a direct and inexpensive analysis completed by a programmable pocket calculator. The objective of this interpretation program is to obtain a fine-scale correlation between discrete temperature logs and lithology. Since the plastic or steel well casings have no observable effect on the temperature gradient profile (CONAWAY & BECK, 1977), the temperature logs combined with gamma logs can be used to find the lithology and to make stratigraphic correlations in areas with wells of which no wire-line logs are available. Limitations posed upon this method are; (1) the well should be in thermal equilibrium with the surrounding formations, (2) the thermal transport by groundwater flow has to be very small (KAPPELMEYER & HAENEL, 1974), and (3) the well diameter should be small to avoid convective heat transport (SAMMEL, 1968).

[1] Manuscript received: 1981-12-29.
Revised manuscript accepted: 1982-02-05.
[2] Institute of Earthsciences, Free University, De Boelelaan 1085, 1081 HV Amsterdam, The Netherlands.

PROGRAM OPERATION

Input data

Temperatures at equidistant intervals (in °C), depth and depth intervals (in metres, only one decimal allowed) are needed.

Program

Through three points or more a straight line is fitted using the least square method (HP: STAT PAC). This line is adjusted each time a new point is entered. If this new point deviates from its estimated value by more than a pre-set value, it is omitted in the calculation. This is done to allow for measurement errors. When two consecutive points deviate, the program assumes these two to belong to a new layer. Table I shows the program listing.

Operation

After the program is entered into the HP-41C calculator, it can be initialized by XEQ 'TL'. From there on the calculator

Table I
Program for the analysis of temperature log data (applicable using a HP – 41C pocket calculator)

```
 01♦LBL "TL"
 02♦LBL A
CLRG  .00201  STO 01
"LAST D? (M)"  PROMPT
10  *  STO 00  CF 04
"D STEP? (M)"  PROMPT
1 E4  /  ST+ 00  SF 01
ΣREG 10

 19♦LBL 07
RCL 00  INT  10  /
ENTER↑  FIX 1  "T "
ARCL X  "⊢ M ?"  PROMPT
FC? 01  GTO 02  X<>Y  Σ+

 34♦LBL 10
DSE 00  ISG 01  GTO 07
GTO 11

 39♦LBL 06
RCL 00  INT  10  /
RCL 05  *  RCL 06  +
STO 02  CF 01  GTO 07

 51♦LBL 02
STO 09  RCL 02  -
STO 16  ST+ 17  ABS  .01
X>Y?  GTO 06  RCL 16
ST- 17  FS? 04  GTO 03
"DEVIATION"  AVIEW
SF 04  GTO 10

 69♦LBL 06
CF 04  RCL 00  INT  10
/  RCL 09  X<>Y  Σ+
GTO 10

 79♦LBL 03
RCL 00  INT  10  /
STO 01  RCL 00  FRC
2 E4  *  RCL 01  +
"TOP="  ARCL X  AVIEW
STOP  RCL 05  100  *
FIX 2  "∡="  ARCL X
"⊢ C/100M"  AVIEW  STOP
RCL 08  SQRT  FIX 4
"R="  ARCL X  AVIEW
CF 04  STOP  RCL 17
X≠0?  GTO 12  "NO FLOW"
AVIEW  GTO A

 118♦LBL 12
0  X>Y?  GTO 13
"FLOW: UPWARD"  AVIEW
GTO A

 125♦LBL 13
"FLOW:DOWNWARD"  AVIEW
GTO A

 129♦LBL 11
RCL 15  RCL 11  RCL 10
RCL 10  XEQ 09  STO 03
RCL 12  RCL 11  RCL 10
RCL 14  XEQ 09  RCL 03
/  STO 04  STO 06
RCL 15  RCL 14  RCL 10
RCL 12  XEQ 09  RCL 03
/  STO 05  RCL 04
RCL 12  *  RCL 05
RCL 14  *  +  RCL 12
X↑2  RCL 15  /  STO 09
-  RCL 13  RCL 09  -  /
STO 08  GTO 06

 172♦LBL 09
*  STO 07  RDN  *
RCL 07  -  RTN
180 END
```

will ask with alpha messages the data to be entered. First the deepest measurement point and the measurement interval are asked by 'LAST D? (M)' and 'D STEP? (M)'. After each value is entered R/S is pressed. Consequently the temperatures at decreasing depths will be asked starting with 'T 200.0M?' if 200 m is the lowest measurement point. If the measurement interval is 2.5 m the next question will be 'T 197.5M?'. The entered temperatures will be compared with the estimated temperatures according to the regression line. If the difference is greater than 0.01°C 'DEVIATION' is displayed and the input can be continued. When the difference in temperature between the estimated one and the next point is also more than 0.01°C, the end of this layer is reached. The decision value of 0.01°C depends on the measurement accuracy. Satisfactory results are obtained using two times the relative accuracy (0.005°C in our case), so the value of 0.01 is assigned in step 58 of the program (KEET, 1981).

Output

After the second deviating point is entered, the top of the layer is displayed by for example 'TOP= 190.0'.After R/S is pressed the temperature gradient is shown by' = 1.52 C/100M' in °C per 100 m. Pressing the R/S a second time reveals the coefficient of correlation (HP: STAT PAC) indicative for the goodness of fit of the line. The final press on R/S shows whether the flow is up, down, or zero. This can be used to check limitation 2. If the flow is in the same direction in several layers, one should be careful with the results. Finally, all registers are cleared automatically and the next layer on top of the last one can be analyzed.

RESULTS AND RECOMMENDATIONS

The interpreted temperature logs are compared with the existing logs of serveral observation boreholes in Friesland, The Netherlands (KEET, 1981). Figure 1 shows part of a log. Although the measurement interval is 2.5 m, the correlation is significant. The accuracy of the relative temperature measurements is 0.005°C. Note the distinct gradient if fine sand is present in a coarse sand formation (as between 123 – 131 m, 142.5 – 152.5 m and 171 – 175 m), while the other logs show no pronounced changes. It can be seen that low gradients occur in coarse sediments. The layers with moderately high or high gradients can be further distinguished with the gammalog. Obviously smaller intervals will lead to still more accurate results.

CONCLUSIONS

The pocket calculator program discribed can be used to give additional information to existing logs. It enables the geologist to gain information from boreholes if no logs were made before they were cased, especially if it is combined with nuclear logs like the gamma log. Assuming the thermal gradient of a formation is constant for some areal extent, it can also be used as a tool in stratigraphic correlation.

ACKNOWLEDGEMENT

I am indebted to E. Seyhan and P. de Beukelaar for reviewing the text.



Fig. 1
Comparision of several logs with the temperature log and the program output: the temperature gradient log

REFERENCES

Conaway, J. G. & A. E. Beck 1977 Fine-scale correlation between temperature gradient logs and lithology – Geophysics 42: 1401-1410.

HP – 41C STAT PAC, Hewlett Packard Corp. (Corvallis, Ore.) 00041-90030 Rev. B 1/80: 32-35.

Kappelmeyer, O. & R. Haenel 1974 Geothermics, with Special Reference to Application – Geoexplo. Monogr., series 1-4, (Berlin): 158 pp.

Keet, B. 1981 Temperature logs and their correlation with excisting logs and surface geophysical measurements near Joure, Friesland, The Netherlands – Int. Rap. Inst. for Earth Sc. Free University, Amsterdam, The Netherlands: 29 pp.

Sammel, E. A. 1968 Convective flow and its effect on temperature logging in small-diameter wells – Geophysics 33: 1004-1012.